

Escape the Desktop with ODS EPUB

David W. Kelley, SAS Institute Inc., Cary, NC

ABSTRACT

The Base SAS® 9.4 Output Delivery System (ODS) EPUB destination enables users to deliver SAS® reports as e-books on Apple mobile devices. ODS EPUB e-books are truly mobile – you don't need an Internet connection to read them. Just install Apple's free [iBooks](#) app, and you're good to go.

This paper shows you how to create an e-book with ODS EPUB and sideload it onto your Apple device. You will learn new SAS 9.4 techniques for including text, images, audio, and video in your ODS EPUB e-books. You will understand how to customize your e-book's table of contents (TOC) so that readers can easily navigate the e-book. And you will learn how to modify the ODS EPUB style to create specialized presentation effects.

This paper provides beginning to intermediate instruction for writing e-books with ODS EPUB. Please bring your iPad, iPhone, or iPod to the presentation so that you can download and read the examples.

INTRODUCTION

Traditionally SAS users have viewed ODS reports on desktop computers. You browse ODS HTML output with Internet Explorer, Firefox, or Chrome. You view print-ready ODS PDF output with Adobe Reader or Acrobat. You run Microsoft PowerPoint to review the slide presentation you created with ODS POWERPOINT (SAS 9.4 and later). And you run Microsoft Excel to peruse the spreadsheet you made with ODS EXCEL (third maintenance release of SAS 9.4 and later).

That's all well and good, but if you're like many folks, your daily computing extends beyond the desktop. You carry a smartphone or tablet to browse the web, check email, and send messages to friends and colleagues. Why not view ODS reports on your mobile device? Now you can, if your device is an iPad, iPhone, or iPod!

In SAS 9.4, ODS EPUB enables users to deliver SAS reports as e-books on Apple mobile devices. The first maintenance release of SAS 9.4 adds ODS EPUB3, which offers powerful new multimedia and presentation features to report writers. For the third maintenance release of SAS 9.4, ODS EPUB is aliased to ODS EPUB3.

This paper shows you how to escape the desktop with ODS EPUB. "[Drive with iCloud, Read with iBooks](#)" demonstrates creating an EPUB e-book with ODS EPUB and sideloading it to iBooks. "[Letters from Caroline](#)" introduces my ODS EPUB project, which involves digitizing letters. "[A Day in the Life](#)" shows you how to easily add paragraph and list text to an ODS EPUB e-book. "[Picture Perfect](#)" teaches you how to combine text and images in the e-book. And "[Then and Now](#)" goes full-multimedia with examples of e-book audio and video.

Now that we've plotted our escape, let's get started!

DRIVE WITH ICLOUD, READ WITH IBOOKS

ODS EPUB e-books are built for mobility. They are created in the [EPUB](#) (Electronic Publication) format, which is a free open-standard format. EPUB is based on HTML 5, which supports Scalable Vector Graphics (SVG) for crisp-looking graphics on any form-factor device. EPUB text is reflowable, meaning that it automatically resizes based on the device screen size and orientation. And EPUB e-books are designed to be fully self-contained. Once you sideload your ODS EPUB report onto your device, you can read it with or without an Internet connection. All you need are sufficient battery charge and the free Apple iBooks app. You can download and install iBooks from the App Store.

You have several options for sideloading ODS EPUB e-books to iBooks. My SAS Global Forum paper (Kelley, Langston, and Summers, 2013) discusses three techniques; see the "[References](#)" section for more information. Since that time, Apple's [iCloud Drive](#) has emerged as an attractive fourth option.

iCloud Drive is a cloud-based document storage management system. If you're familiar with Dropbox or Google Docs, then you'll understand iCloud Drive right away. iCloud Drive clients run on Apple devices and Windows computers. A web-based interface, icloud.com, is also available.

The iCloud Drive app comes with iOS 9 and iOS 10. To turn on the iCloud Drive app in iOS 9:

1. Launch the **Settings** app from your Home Screen.
2. Tap on **iCloud**.
3. Tap on **iCloud Drive**.
4. Switch **Show on Home Screen** to **On**.

The instructions for iOS 10 are the same as for iOS 9 except that you omit step 4.

I installed the free Windows iCloud Drive client on my work PC. You can download it from <https://support.apple.com>.

The iCloud Windows installer created a special directory on my PC: C:\Users\MyUserName\iCloudDrive. Any directories or files I store in this directory are automatically synchronized to iCloud Drive "in the cloud" and thus become available to all of my iCloud Drive clients.

If you're an ODS veteran, then you probably already know how I sideloaded ODS EPUB e-books onto my iPad – I simply started a Windows version of SAS and executed code like this:

```
ods graphics on;
ods epub file="c:\users\MyUserName\icloudrive\freq.epub" title="Freq Test"
;
proc freq data=sashelp.class;
tables sex*age/plots=freq;
run;
ods epub close;
```

The ODS GRAPHICS statement enables the generation of graphics from statistical procedures. In this case the FREQ procedure generates an optional frequency plot after the standard crosstabulation report.

ODS EPUB is one of many supported ODS destinations. They all use the "ODS sandwich" programming idiom. You open one or more destinations, run your procedure and DATA steps, and then close the destination(s). Destinations remain open until explicitly closed or the SAS session ends.

The FILE= option specifies the file to receive the generated e-book. The default filename is "sasepub.epub". EPUB e-books are required to have an "epub" extension.

The TITLE option specifies the required e-book title, which defaults to "SAS Output". You should always specify a title for your ODS EPUB e-book.

After executing the SAS code, I ran the iCloud Drive app on my iOS 9 iPad. The app displayed the list of available documents, as is seen in Figure 1.




Figure 1: iCloud Drive List View

I tapped **freq.epub**, and then requested the app to download the e-book from the cloud to my iPad. Once downloaded, I exported it to iBooks by tapping first the export icon and then **Copy to iBooks**, as is shown in Figure 2.




Figure 2: Export EPUB E-book to iBooks

The iBooks app launched and displayed the e-book's PROC FREQ output; see Figure 3.




Figure 3: iBooks Display of PROC FREQ Output

The iCloud Drive app user experience is similar with iOS 10, except that the app doesn't require you to explicitly download the e-book. The app implicitly downloads the e-book when you export it to iBooks.

Soup to nuts! Now you know what ODS EPUB is and how to use it to make EPUB e-books for your Apple mobile device. But why would you want to do that? I can only imagine, but I know why *I* wanted to use it for one of *my* projects. I'll explain that in the next section.

LETTERS FROM CAROLINE

My mother attended [Meredith College](#) in Raleigh, NC from 1952 to 1955. During that time she regularly exchanged letters with her parents in Durham, NC. My grandmother saved my mother's letters, and eventually they passed to me.

In 2013 I decided to digitize the letters. My idea was to virtually bind them into an ODS EPUB e-book, where each letter would be a chapter in the e-book. The chapters would be ordered chronologically by letter date.

At the time I used the first maintenance release of SAS 9.4 to implement my ODS EPUB e-book project. Since then I've updated the SAS code as new ODS EPUB features have appeared in later versions, up to and including the fourth maintenance release of SAS 9.4.

Here is the top slice of the "ODS sandwich" code to create my e-book:

```
ods epub file="letters.epub" title="Letters from Caroline"
  style=letterStyle
  options(creator="David W. Kelley"
    subject="NON-CLASSIFIABLE"
    cover_image="media/Caroline_cover.jpg");
```

Here and elsewhere, I've edited the code for purposes of clarity and brevity.

The STYLE option tells ODS EPUB what style to use. If you don't specify a style, then ODS EPUB uses its default style, Daisy. The SAS code tells ODS EPUB to use the letterStyle style, which I developed specifically for digitizing my mother's letters. We'll discuss the letterStyle style in the next section.

The OPTIONS option allows you to control various ODS EPUB e-book behaviors and to specify metadata for the e-book. The CREATOR suboption specifies the e-book's creator (author). For better or worse, I'm the author of my e-book, so my name goes there. By default an ODS EPUB e-book has no author.

The SUBJECT suboption identifies the e-book's subject. I struggled to come up with an appropriate subject for 'Letters from Caroline'. Is it biography, history, or something else? According to the [Book Industry Study Group](#), you should specify the subject category NON-CLASSIFIABLE if no other category fits, so that's what I did. By default an ODS EPUB e-book has no subject.

iBooks displays the author metadata in the Library and TOC views. It displays the subject metadata in the Library view. If you supply a cover image with the COVER_IMAGE suboption, then iBooks displays the image in the Library and TOC views. For best results, your cover image should be in JPEG or PNG format, with a minimum DPI of 300 and a length along the shorter axis of at least 1400 pixels. By default an ODS EPUB e-book has no cover image.

Figure 4 shows how iBooks displays the author metadata and cover image for 'Letters from Caroline' in the TOC view.




Figure 4: iBooks TOC View for 'Letters from Caroline'

A DAY IN THE LIFE

While attending Meredith, my mother wrote 100 letters to my grandparents. The first letter is dated September 11, 1952, and the last letter is dated May 18, 1955.

In her letter dated May 27, 1953, she discusses end-of-year activities. Figure 5 shows how iBooks displays the letter.




Figure 5: iBooks View of May 27, 1953 Letter

For this and the other letters, I transcribed the text into SAS code. The ODSTEXT procedure, which was introduced in SAS 9.4, made this task straightforward. PROC ODSTEXT knows how to deal with headings, paragraphs, and lists, which account for 99% of the content of the letters. In fact, for the whole project I used only one other procedure besides PROC ODSTEXT, and that was the TEMPLATE procedure, for my custom letterStyle style.

Here is the PROC TEMPLATE code for the letterStyle style:

```
proc template;
define style styles.letterStyle;
parent=styles.daisy;

class letterDate /
  just = r
;
class greeting /
  textindent = 0
;
class paragraph /
  textindent = 2em
;
class fonts "Fonts used in the default style" /
  "DocFont" = ("

```

```

"BatchFixedFont" = ("SAS Monospace, <monospace>,
                    Courier, monospace", 9pt )
"FixedHeadingFont" = ("<monospace>, Courier,
                       monospace", 8pt )
"FixedStrongFont" = ("<monospace>, Courier,
                      monospace", 8pt, bold)
"FixedEmphasisFont" = ("<monospace>, Courier,
                        monospace", 8pt, italic)
"EmphasisFont" = ("<serif>, Times New Roman,
                  serif", 10pt, italic)
"StrongFont" = ("<serif>, Times New Roman,
                serif", 10pt, bold)
"TitleFont" = ("<serif>, Times New Roman,
               serif", 14pt, bold)
"TitleFont2" = ("<serif>, Times New Roman,
                serif", 12pt, bold)
;
end;
run;

```

I designed the letterStyle style so that PROC ODSTEXT text could be presented in a way that mimicked my mother's letter-writing style:

- I use a serif typeface. My mother wrote some of her letters on a typewriter, and the serif typeface more closely resembles the slab-serif typewriter typeface than does Daisy's sans-serif typeface. Later I experimented with an actual typewriter (monospace) typeface, but it looked horrible in iBooks, so I stuck with the serif typeface.
- I right-justify the date text (letterDate style).
- I indent the paragraph text (paragraph style) but not the greeting text (greeting style).

Here is a snippet of the SAS code for the letter:

```

ods epub newchapter=now;
ods proclabel="May 27, 1953";

proc odstext contents="";
p "Wednesday afternoon"/style=letterDate;

p "Hello!"/style=greeting;

p "Well, I'm trying to get around to my ""weekly""
  letter, but I think that I've done pretty well
  this week because I managed to write an extra
  card. I'm trying to make up for next week when
  I won't be able to do anything but study. I've
  found out most of my plans, so I'll relay them
  to you.";

p "I talked to Mrs. Mayes, the house director,
  today, and she said they would be glad to store
  my desk and anything else I wanted stored for
  the summer. I know that Daddy will be glad to

```

```
hear that. As far as I know now, the thing for
you to do is to come over here as early as you
can Friday afternoon and pack my clothes. I'm
not going to try to pack anything.";
```

```
p "My exam schedule:";
```

```
list/style={liststyletype="none"};
  item "Saturday: Health Education";
  item "Monday: Biology";
  item "Wed: English";
  item "Thurs: History";
  item "Fri: Spanish & Religion";
end;
run;
```

Remember that I want each letter to have its own chapter. ODS EPUB NEWCHAPTER=NOW tells ODS EPUB to begin a new chapter in the e-book. By default ODS EPUB writes output to a single chapter.

The ODS PROCLABEL statement specifies the text to be displayed in the TOC for the PROC ODSTEXT step. I set the text to be the letter date. The default PROC ODSTEXT TOC text ("The ODSText Procedure") is unhelpfully generic and should always be customized. By default PROC ODSTEXT puts a second bit of text ("TextBlock") in the TOC, and it is similarly unhelpful. I suppress it by specifying CONTENTS="" on the PROC ODSTEXT statement.

As you can see from the code, the PROC ODSTEXT P statement is for paragraphs and the LIST statement is for lists. (I told you that the code was straightforward. 😊) Each statement accepts a STYLE option for customizing that statement's text. That's how we get the desired non-default text indentation and justifications. I also use the STYLE option to suppress list item markers (bullets) for the day items in the exam schedule list.

The May 27, 1953 letter has more or less the same format as the other 99 letters. Collectively these letters comprise the main content in 'Letters from Caroline'. I created additional sections for the e-book to help put that content into context and to give the e-book a more book-like structure. One such section is the title page, which we'll consider next.

PICTURE PERFECT

After I finished transcribing my mother's letters into 'Letters from Caroline' chapters, I decided to add a title page to my e-book. Strictly speaking, this wasn't required, because iBooks automatically displays the title and author metadata, as well as the cover image. However, I spot-check ODS EPUB e-books with other popular EPUB e-readers, and their automatic support for this metadata varies. Although ODS EPUB e-books are optimized for iBooks, they should work acceptably well with any quality EPUB e-reader.

Figure 6 shows how iBooks displays the title page of 'Letters from Caroline'.




Figure 6: iBooks View of 'Letters from Caroline' Title Page

The SAS code for the title page follows:

```
ods epub newchapter=now options(chapter_type="frontmatter titlepage");
ods escapechar='^';
ods proclabel="Letters from Caroline";

proc odstext contents="";
h1 "Letters from Caroline"/style={just=c};
p "^{image media/Caroline1.jpg?
    width=50%;alt_desc=Caroline Knight Garrett}"
  /style={textindent=0 just=c};
p "David W. Kelley"/style={textindent=0 just=c};
run;
```

ODS EPUB NEWCHAPTER=NOW starts a new chapter in the e-book. The CHAPTER_TYPE option indicates what type of chapter it is. The semantic terms "frontmatter" and "titlepage" are taken from the [EPUB 3 Structural Semantics Vocabulary](#). The default semantic terms for a chapter are "bodymatter" and "chapter". Assigning semantics to chapters makes it possible for EPUB e-book readers to provide context-sensitive behaviors for an e-book. The bad news is that iBooks has limited support for semantics, but the good news is that ODS EPUB generates the semantics most useful to iBooks by default.

The ODS ESCAPECHAR option defines a character that will trigger inline text processing. The ODS PROCLABEL option specifies the text displayed in the TOC for the PROC ODSTEXT step.

PROC ODSTEXT is the one-stop procedure for my title page content. I use the H1 statement, new in the fourth maintenance release of SAS 9.4, to define the title text. Like HTML, PROC ODSTEXT supports six levels of headings, H1-H6. By default headings are left-justified, so I specify STYLE= to override with center justification.

The first P statement defines no text. Instead, it serves as a convenient hook for an ODS EPUB IMAGE inline function that stores the supplied image in the e-book. I optionally specify a width of 50%, which means that

iBooks will render the image such that it occupies 50% of the width of the title page. Specifying the width as a percentage allows iBooks to behave consistently in portrait or landscape orientation with any Apple mobile device. The optional alternate description offers a way for the iOS VoiceOver accessibility feature to describe the image to a reader with low or no vision. The STYLE= specifies center text justification and no text indentation. This overrides the default paragraph text justification (left) and indentation (2em).

The title page image is in JPEG format; ODS EPUB also supports the GIF, PNG, and SVG image formats.

The second P statement names the author; the STYLE= puts the name in the right place.

You might wonder why I use an inline function for the image. That's because ODS doesn't currently offer a general-purpose way to include an external image in generated output. The IMAGE inline function is unique to ODS EPUB. Don't try it with ODS PDF or ODS HTML!

This example demonstrates that PROC ODSTEXT is good for more than just text and lists. With PROC ODSTEXT, you can easily combine ODS EPUB text with media such as images. In the next section we'll learn how to integrate ODS EPUB text with video and audio.

THEN AND NOW

My mother left Meredith College in 1955 to get married and move to Germany. She continued her education by correspondence (of course!) and graduated in 1957.

My daughter Jane started at Meredith in 2015. Although much has changed on campus in the 60 years between my mother's departure and my daughter's arrival, some things remain the same. For example, the dormitory my mother lived in her freshman year, Stringfield, still houses students.

Meredith celebrated its 125th anniversary in 2016. As a part of that year's Founders' Week activities, the college released a video with scenes from its history. I decided to include the video in my e-book. That way readers could get another perspective on the place where my mother wrote her letters.

Here is the SAS code for the video:

```
ods epub newchapter=now
      options(chapter_type="backmatter appendix");
ods proclabel=
  "Appendix C: Meredith College at 125 Years";

proc odstext contents="";
h1 "Meredith College at 125 Years";
p "^{video media\MC125GoingStrong.mp4?
  controls=controls;
  poster=media\MCFirstCampus.jpg;
  height=150px;width=300px}";
p "Video released during Founders' Week, 2016";
run;
```

ODS EPUB NEWCHAPTER=NOW starts a new chapter in the e-book. The CHAPTER_TYPE option indicates that the chapter content functions as an appendix. The ODS PROCLABEL option specifies the text displayed in the TOC for the chapter.

Once again PROC ODSTEXT is our workhorse procedure. The H1 statement specifies the heading text. The first P statement defines no text, but rather serves as an integration point for the ODS EPUB VIDEO inline function, which stores the given video in the e-book. The video is in MP4 format; ODS EPUB also supports the M4V and MOV video formats.

The VIDEO inline function is based on the HTML5 [video](#) tag. The CONTROLS option indicates that iBooks should surface its native video control. The required POSTER option specifies a poster image to be displayed prior to playback. The HEIGHT and WIDTH options determine the dimensions of the control.

The second P statement functions like a caption, providing useful context for the video.

Figure 7 shows how iBooks displays the video control.




Figure 7: iBooks Display of Meredith Video

My mother and daughter have a Meredith connection, and I wanted to recognize that connection in 'Letters from Caroline'. I asked Jane to choose a letter, and we made an audio recording of her reading it. I included the audio as an appendix in the e-book. The SAS code for the audio follows:

```
ods epub newchapter=now
      options(chapter_type="backmatter appendix");
ods proclabel=
  "Appendix D: Jane Kelley Reads September 11, 1952 Letter";
```

```
proc odstext contents="";
h "Jane Kelley Reads September 11, 1952 Letter";
p "Jane Kelley is a member of the Meredith College
  Class of 2019. Listen to her read the
  ^{style [url='chapter11.html']letter} that her
  grandmother, Caroline Kelley, wrote on
  September 11, 1952.";
p "^{audio media\CarolineLetter.m4a?controls=controls}";
run;
```

The audio code is mostly similar to the just-discussed video code, so I'll highlight the differences. The first PROC ODSTEXT P statement defines a hyperlink to the letter Jane reads in the audio. (Recall that each letter is its own chapter.) To learn more about how ODS EPUB maps e-book chapters to internal HTML filenames, see my SAS Global Forum paper (Kelley, 2014) in the "[References](#)" section.

The second P statement uses the AUDIO inline function, which is patterned after the HTML5 [audio](#) tag. The audio is in M4A format; ODS EPUB also supports the MP3, MP4, and WAV audio formats.

Figure 8 shows how iBooks displays the audio control.

Jane Kelley Reads September 11, 1952 Letter

Jane Kelley is a member of the Meredith College Class of 2019. Listen to her read the [letter](#) that her grandmother, Caroline Kelley, wrote on September 11, 1952.




Figure 8: iBooks Display of Audio Control

In the September 11, 1952 letter, my mother complains about the difficulty of freshman placement tests, talks about staying up "until the wee hours of the morning" getting to know her new roommate, and looks forward to social events, especially the ones with college males. (Meredith is a women's college.) Some things never change!

CONCLUSION

You've learned how to create fully mobile ODS reports with ODS EPUB. You've seen that the text, image, and multimedia capabilities in ODS EPUB extend the boundaries of SAS reporting beyond the traditional tables and graphs. I've explained why and how I used ODS EPUB for one of my projects. Now it's your turn. What new ODS EPUB e-book do you want to create, and why?

Now that you're prepared, come up with an idea, formulate your plan, and then make the escape from the desktop!

REFERENCES

Kelley, David W.; Langston, Julianna; Summers, Ed. 2013. "Go Mobile with the ODS EPUB Destination." *Proceedings of the SAS Global Forum 2013 Conference*. Cary, NC: SAS Institute Inc. Available at <http://support.sas.com/resources/papers/proceedings13/368-2013.pdf>.

Kelley, David W. 2014. "Advanced Mobile Reporting with the ODS EPUB3 Destination." *Proceedings of the SAS Global Forum 2014 Conference*. Cary, NC: SAS Institute Inc. Available at <http://support.sas.com/resources/papers/proceedings14/SAS339-2014.pdf>.

ACKNOWLEDGMENTS

The author would like to thank SAS employee Allison Crutchfield for carefully reviewing the paper.

RECOMMENDED READING

Berjon, Robin, et al. "HTML5." *W3C*. October 28, 2014. Available at <http://www.w3.org/TR/html5/>.

"EPUB 3 Structural Semantics Vocabulary." *International Digital Publishing Forum*. January 5, 2016. Available at <http://www.idpf.org/epub/vocab/structure/>.

Conboy, Garth, et al. "EPUB 3.1 Overview." *International Digital Publishing Forum*. January 5, 2017. Available at <http://www.idpf.org/epub/31/spec/epub-overview.html>.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author:

David W. Kelley
SAS Institute Inc.
100 SAS Campus Drive
Cary, NC 27513-2414, USA
David.Kelley@sas.com
<http://www.sas.com>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.